
**************************************************************
******************** Replication code for ********************
***** "Do International Rankings Affect Public Opinion?" *****
*** British Journal of Politics and International Relations **
************ Asif Efrat, Amnon Cavari, Omer Yair *************
**************************************************************


* Setting the working directory
cd <<<"ADD DIRECTORY HERE">>>

* Upload the .dta dataset
use "BJPIR_ranking_dataset.dta", clear



******************************
*** Descriptive statistics ***
******************************

sum Q21_age 
tab Q22_gender
tab ideo_bloc


*** The DVs ***
* First DV: evaluations
tab Q3_DV1_ranking
recode Q3_DV1_ranking (6=.), gen (Q3_01) /* Removing DK responses */
replace Q3_01=(Q3_01-1)/4
sum Q3_01
ttest Q3_01, by (HR_exp_ENV) /* p-value<.001 */

* Second DV: attitudes
tab Q4_DV2_ranking
recode Q4_DV2_ranking (6=.), gen (Q4_01) /* Removing DK responses */
replace Q4_01=(Q4_01-1)/4
sum Q4_01
ttest Q4_01, by (HR_exp_ENV) /* p-value<.001 */


* Third DV: behaviroal intentions
tab Q7_DV3_ranking
recode Q7_DV3_ranking (6=.), gen (Q7_01) /* Removing DK responses */
replace Q7_01=(Q7_01-1)/4
sum Q7_01
ttest Q7_01, by (HR_exp_ENV) /* p-value<.001 */



****************
*** Figure 1 ***
****************

* First DV
quietly reg Q3_01 HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, drop(_cons) xline(0) level(90) graphregion(color(white)) ytitle("") ///
title("Model 1. Good performance", size(medium)) ///
fxsize(90) coeflabel(HR_exp_rank_high = "High rank" ///
			       1.age_groups = "Ages 18-29" ///
			       3.age_groups = "Ages 50-64" ///
			       4.age_groups = "Ages 65-99" ///
				   1.college_educ = "College" ///
				   Q25_religiosity = "Religiosity")
graph save "Graph1.gph", replace

* Second DV
quietly reg Q4_01 HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, drop(_cons) xline(0) level(90) graphregion(color(white)) ytitle("") ///
title("Model 2. More effort", size(medium)) ///
fxsize(65) coeflabel(HR_exp_rank_high = " " ///
			       1.age_groups = " " ///
				   3.age_groups = " " ///
			       4.age_groups = " " ///
				   1.Q22_gender = " " ///
				   1.college_educ = " " ///
				   1.jewish = " " ///
				   Q25_religiosity = " " ///
				   1.ideo_bloc = " " ///
				   2.ideo_bloc = " ")
graph save "Graph2.gph", replace

* Third DV
quietly reg Q7_01 HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, drop(_cons) xline(0) level(90) graphregion(color(white)) ///
ytitle("") title("Model 3. Sign petition", size(medium)) ///
fxsize(65) coeflabel(HR_exp_rank_high = " " ///
			       1.age_groups = " " ///
				   3.age_groups = " " ///
			       4.age_groups = " " ///
				   1.Q22_gender = " " ///
				   1.college_educ = " " ///
				   1.jewish = " " ///
				   Q25_religiosity = " " ///
				   1.ideo_bloc = " " ///
				   2.ideo_bloc = " ")
graph save "Graph3.gph", replace

** Figure 1 - Combining table1a, table1b, and table1c **
graph combine Graph1.gph Graph2.gph Graph3.gph, ///
ycommon xcommon row(1) graphregion(color(white)) 

erase Graph1.gph 
erase Graph2.gph
erase Graph3.gph



************************************************
** Figure 2 estimates & pretreatment analyses **
************************************************

** The two pre-treatmennt items **
* HR evaluation (satisfaction with Israel's HR situation)
tab Q2_SATIS_HUMANRIGHTS 
recode Q2_SATIS_HUMANRIGHTS (6=.), gen (Q2_pre_HR) /* Removing DK responses */
replace Q2_pre_HR=(Q2_pre_HR-1)/4
sum Q2_pre_HR
pwcorr Q2_pre_HR Q3_01 if HR_exp_ENV==0, obs sig
* Balance tests:
ttest Q2_pre_HR, by (HR_exp_ENV) /* p-value=.484 */
ttest Q2_pre_HR if HR_exp_ENV==0, by (HR_exp_rank_high) /* p-value=.652 */


* Environment evaluations (satisfaction with Israel's Environment situation)
tab Q2_SATIS_ENV
recode Q2_SATIS_ENV (6=.), gen (Q2_pre_ENV) /* Removing DK responses */
replace Q2_pre_ENV=(Q2_pre_ENV-1)/4
sum Q2_pre_ENV
pwcorr Q2_pre_ENV Q3_01 if HR_exp_ENV==1, obs sig
* Balance tests: (ONE FAILURE...)
ttest Q2_pre_ENV, by (HR_exp_ENV) /* p-value=.443 */
ttest Q2_pre_ENV if HR_exp_ENV==1, by (HR_exp_rank_high) /* p-value=.049 */


* Pre-treatment estimate:
sum Q2_pre_HR if HR_exp_ENV==0  /*(M=0.48)*/
sum Q2_pre_HR if HR_exp_ENV==0 & Q3_01!=. /*(M=0.48)*/
ttest Q2_pre_HR, by (HR_exp_ENV) /* p-value=.484 */

* Experimental estimate:
ttest Q3_01 if HR_exp_ENV==0, by (HR_exp_rank_high) /* p-value<.001 */

* "Comparisons"
ttest Q2_pre_HR=Q3_01 if HR_exp_ENV==0 & HR_exp_rank_high==0
ttest Q2_pre_HR=Q3_01 if HR_exp_ENV==0 & HR_exp_rank_high==1

* In the US source:
ttest Q2_pre_HR=Q3_01 if HR_exp_ENV==0 & HR_exp_rank_high==0 & HR_exp_group==1

* In the NGO source:
ttest Q2_pre_HR=Q3_01 if HR_exp_ENV==0 & HR_exp_rank_high==0 & HR_exp_group==0

* In the OECD source:
ttest Q2_pre_HR=Q3_01 if HR_exp_ENV==0 & HR_exp_rank_high==0 & HR_exp_group==2

* Testing for statistically different "backlash effect"
gen diff_DV1=Q3_01-Q2_pre_HR if HR_exp_ENV==0
oneway diff_DV1 HR_exp_group if HR_exp_ENV==0 & HR_exp_rank_high==0, t sch


* Pre-treatment estimate:
sum Q2_pre_ENV if HR_exp_ENV==1 /*(M=0.40)*/
sum Q2_pre_ENV if HR_exp_ENV==1 & Q3_01!=. /*(M=0.48)*/
ttest Q2_pre_ENV, by (HR_exp_ENV) /* p-value=.443 */

* Experimental estimate:
ttest Q3_01 if HR_exp_ENV==1, by (HR_exp_rank_high) /* p-value<.001 */

* "Comparisons"
ttest Q2_pre_ENV=Q3_01 if HR_exp_ENV==1 & HR_exp_rank_high==0
ttest Q2_pre_ENV=Q3_01 if HR_exp_ENV==1 & HR_exp_rank_high==1



****************
*** Figure 3 ***
****************

* First DV
gen inter=HR_exp_rank_high*HR_exp_ENV 

quietly reg Q3_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_ENV inter) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 1. Good performance", size(medium)) ///
fxsize(90) coeflabel(HR_exp_rank_high = "High rank" ///
			       HR_exp_ENV = "Environment" ///
				   inter = "High X Env.")
graph save "Graph1.gph", replace

reg Q3_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
lincom HR_exp_rank_high /* The high-ranking slope in the Human Rights condition*/
lincom HR_exp_rank_high + inter /* The high-ranking slope in the Environment condition*/

* Second DV
quietly reg Q4_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_ENV inter) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 2. More effort", size(medium)) ///
fxsize(70) coeflabel(HR_exp_rank_high = " " ///
			       HR_exp_ENV = " " ///
				   inter = " ")
graph save "Graph2.gph", replace

reg Q4_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
lincom HR_exp_rank_high  /* The high-ranking slope in the Human Rights condition*/
lincom HR_exp_rank_high + inter /* The high-ranking slope in the Environment condition*/

* Third DV
quietly reg Q7_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_ENV inter) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 3. Sign petition", size(medium))   ///
fxsize(70) coeflabel(HR_exp_rank_high = " " ///
			       HR_exp_ENV = " " ///
				   inter = " ")
graph save "Graph3.gph", replace

reg Q7_01 HR_exp_rank_high HR_exp_ENV inter ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
lincom HR_exp_rank_high  /* The high-ranking slope in the Human Rights condition*/
lincom HR_exp_rank_high + inter /* The high-ranking slope in the Environment condition*/

** Figure 3 - Combining table2a, table2b, and table2c **
graph combine Graph1.gph Graph2.gph Graph3.gph, ///
ycommon xcommon row(1) graphregion(color(white)) 

erase Graph1.gph 
erase Graph2.gph
erase Graph3.gph



*****************************
*** Figure 4 (new figure) ***
*****************************

* First DV
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
margins i.HR_exp_rank_high#i.HR_exp_ENV
marginsplot, level(90) graphregion(color(white)) ytitle("Predicted values")  ///
xtitle("") title("Model 1. Good performance", size(medium))
graph save "Graph1.gph", replace

* Second DV
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
margins i.HR_exp_rank_high#i.HR_exp_ENV
marginsplot, level(90) graphregion(color(white)) ytitle("Predicted values")  ///
xtitle("") title("Model 2. More effort", size(medium))
graph save "Graph2.gph", replace


* Third DV
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
margins i.HR_exp_rank_high#i.HR_exp_ENV
marginsplot, level(90) graphregion(color(white)) ytitle("Predicted values")  ///
xtitle("") title("Model 3. Sign petition", size(medium)) 
graph save "Graph3.gph", replace

** Figure 4 - Combining table2a, table2b, and table2c **
grc1leg  graph1.gph graph2.gph graph3.gph, ///
ycommon xcommon row(1) graphregion(color(white)) 

erase graph1.gph 
erase graph2.gph
erase graph3.gph



****************
*** Figure 5 ***
****************

* First DV
gen inter2=HR_exp_rank_high*HR_exp_group_US

quietly reg Q3_01 HR_exp_rank_high HR_exp_group_US inter2 ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_group_US inter2) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 1. Good performance", size(medium)) ///
fxsize(90) coeflabel(HR_exp_rank_high = "High rank" ///
			       HR_exp_group_US = "US" ///
				   inter2 = "High X US")
graph save "Graph1.gph", replace
/*mlabel format(%9.2f) mlabposition(12) mlabgap(*2) /// */

* Second DV
quietly reg Q4_01 HR_exp_rank_high HR_exp_group_US inter2 ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_group_US inter2) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 2. More effort", size(medium))  ///
fxsize(70) coeflabel(HR_exp_rank_high = " " ///
			       HR_exp_group_US = " " ///
				   inter2 = " ")
graph save "Graph2.gph", replace

* Third DV
quietly reg Q7_01 HR_exp_rank_high HR_exp_group_US inter2 ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
coefplot, keep(HR_exp_rank_high HR_exp_group_US inter2) xline(0) level(90) ///
graphregion(color(white)) ytitle("") title("Model 3. Sign petition", size(medium)) ///
fxsize(70) coeflabel(HR_exp_rank_high = " " ///
			       HR_exp_group_US = " " ///
				   inter2 = " ")
graph save "Graph3.gph", replace

** Figure 5 - Combining table3a, table3b, and table3c **
graph combine Graph1.gph Graph2.gph Graph3.gph, ///
ycommon xcommon row(1) graphregion(color(white)) 

erase Graph1.gph 
erase Graph2.gph
erase Graph3.gph




***********************
*** Online Appendix ***
***********************


*****************************************
*** Section A: Descriptive statistics ***
*****************************************

***  Demographic/political items ***
sum Q21_age 
tab Q22_gender 
tab jewish 
tab college_educ 
tab ideo_bloc


********************************
*** Section C: balance tests ***
********************************

oneway Q21_age HR_conditions, t
oneway Q21_age HR_exp_rank_high, t
oneway Q21_age HR_exp_ENV, t
oneway Q21_age HR_exp_group, t 

tab Q22_gender HR_conditions, chi col
tab Q22_gender HR_exp_rank_high, chi col
tab Q22_gender HR_exp_ENV, chi col /*significant*/
tab Q22_gender HR_exp_group, chi col

tab college_educ HR_conditions, chi col
tab college_educ HR_exp_rank_high, chi col
tab college_educ HR_exp_ENV, chi col 
tab college_educ HR_exp_group, chi col 

tab jewish HR_conditions, chi col
tab jewish HR_exp_rank_high, chi col
tab jewish HR_exp_ENV, chi col 
tab jewish HR_exp_group, chi col /*significant*/

oneway Q25_religiosity HR_exp_group, t 
oneway Q25_religiosity HR_exp_rank_high, t
oneway Q25_religiosity HR_exp_ENV, t
oneway Q25_religiosity HR_exp_group, t 

oneway Q26_ideology HR_conditions if Q26_ideology!=8, t
oneway Q26_ideology HR_exp_rank_high if Q26_ideology!=8, t
oneway Q26_ideology HR_exp_ENV if Q26_ideology!=8, t
oneway Q26_ideology HR_exp_group if Q26_ideology!=8, t 

mlogit HR_conditions Q21_age Q22_gender college_educ jewish Q25_religiosity Q26_ideology if Q26_ideology!=8
mlogit HR_exp_rank_high Q21_age Q22_gender college_educ jewish Q25_religiosity Q26_ideology if Q26_ideology!=8
mlogit HR_exp_ENV Q21_age Q22_gender college_educ jewish Q25_religiosity Q26_ideology if Q26_ideology!=8
mlogit HR_exp_group Q21_age Q22_gender college_educ jewish Q25_religiosity Q26_ideology if Q26_ideology!=8



********************************************************
*** Section D: Regression tables of the main results ***
********************************************************

** Table D1 - testing H1a, H1b, and H1c (=the equivalent of Figure 1 in the main text) **
* Model 1: Testing H1a (The state of HR/Environment in Israel is good) ** 
reg Q3_01 i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableD1.doc, replace se dec(2) alpha (.02, .1) symbol (**, *)
retrodesign .0859494, se(.0087749) alpha(0.1) df(3738)
di .0859494/.2744538
esize twosample Q3_01, by (HR_exp_rank_high) /* Cohen d = 0.31 */


* Model 2: Testing H1b (Israeli Gov. should do more to improve the HR/Environment situation)
reg Q4_01 i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD1.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
retrodesign -.0272297, se(.0078411) alpha(0.1) df(3778)
di -.0272297/.2476656
esize twosample Q4_01, by (HR_exp_rank_high) /* Cohen d = 0.13 */

* Model 3: Testing H1c (Willingness to sign a petition calling the Israeli government to act to improve HR/Environment)
reg Q7_01 i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD1.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
retrodesign  -.0317561, se(.0097491) alpha(0.1) df(3568)
di -.0317561/.300507
esize twosample Q7_01, by (HR_exp_rank_high) /* Cohen d = 0.12 */


** Table D2 - testing H2 (=the equivalent of Figure 3 in the main text) **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableD2.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 
retrodesign .053206, se(.017114) alpha(0.1) df(3736)
di .0601636/.2744538
di .1133696/.2744538
esize twosample Q3_01 if HR_exp_ENV==0, by (HR_exp_rank_high) /* Cohen d = 0.21 */
esize twosample Q3_01 if HR_exp_ENV==1, by (HR_exp_rank_high) /* Cohen d = 0.42 */

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD2.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
retrodesign -.0287261, se(.0154493) alpha(0.1) df(3776)
di -.0136134/.2476656
di -.0423395/.2476656
esize twosample Q4_01 if HR_exp_ENV==0, by (HR_exp_rank_high) /* Cohen d = 0.08 */
esize twosample Q4_01 if HR_exp_ENV==1, by (HR_exp_rank_high) /* Cohen d = 0.20 */

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD2.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
retrodesign -.033672, se(.019027) alpha(0.1) df(3566)
di -.0160652/.300507
di -.0497372/.300507
esize twosample Q7_01 if HR_exp_ENV==0, by (HR_exp_rank_high) /* Cohen d = 0.06 */
esize twosample Q7_01 if HR_exp_ENV==1, by (HR_exp_rank_high) /* Cohen d = 0.19 */


** Table D3 - testing H3 (=the equivalent of Figure 5 in the main text) **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableD3.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD3.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableD3.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 



***********************************
*** Section E: Robustness tests ***
***********************************

*** Regression tables of the main results - without any control variables ***

** Table E1 - testing H1a, H1b, and H1c **
* Model 1: Testing H1a (The state of HR/Environment in Israel is good) ** 
reg Q3_01 i.HR_exp_rank_high
outreg2 using TableE1.doc, replace se dec(2) alpha (.02, .1) symbol (**, *)

* Model 2: Testing H1b (Israeli Gov. should do more to improve the HR/Environment situation)
reg Q4_01 i.HR_exp_rank_high 	
outreg2 using TableE1.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Testing H1c (Willingness to sign a petition calling the Israeli government to act to improve HR/Environment)
reg Q7_01 i.HR_exp_rank_high 
outreg2 using TableE1.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Table E2 - testing H2 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV 
outreg2 using TableE2.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_ENV 
outreg2 using TableE2.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_ENV 
outreg2 using TableE2.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Table E3 - testing H3 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group_US 
outreg2 using TableE3.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_group_US 
outreg2 using TableE3.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_group_US
outreg2 using TableE3.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 



*** Regression tables of the main results i with Ordered logistic resgression ***
** Testing H1a, H1b, and H1c **
* Model 1: Testing H1a (The state of HR/Environment in Israel is good) ** 
ologit Q3_DV1_ranking i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q3_DV1_ranking!=6
outreg2 using TableE4.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Testing H1b (Israeli Gov. should do more to improve the HR/Environment situation)
ologit Q4_DV2_ranking i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q4_DV2_ranking!=6
outreg2 using TableE4.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Testing H1c (Willingness to sign a petition calling the Israeli government to act to improve HR/Environment)
ologit Q7_DV3_ranking i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q7_DV3_ranking!=6
outreg2 using TableE4.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Testing H2 **
* Model 1: The state of HR/Environment in Israel is good ** 
ologit Q3_DV1_ranking i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q3_DV1_ranking!=6
outreg2 using TableE5.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
ologit Q4_DV2_ranking i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q4_DV2_ranking!=6	
outreg2 using TableE5.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
ologit Q7_DV3_ranking i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q7_DV3_ranking!=6	
outreg2 using TableE5.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Testing H3 **
* Model 1: The state of HR/Environment in Israel is good ** 
ologit Q3_DV1_ranking i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q3_DV1_ranking!=6
outreg2 using TableE6.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
ologit Q4_DV2_ranking i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q4_DV2_ranking!=6	
outreg2 using TableE6.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
ologit Q7_DV3_ranking i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc if Q7_DV3_ranking!=6	
outreg2 using TableE6.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 



*** Regressions - with 'Don't know' responses treated as mid-category responses ***
gen Q3_01_DK=Q3_01
replace Q3_01_DK=0.5 if Q3_DV1_ranking==6
tab Q3_01_DK Q3_01, miss

gen Q4_01_DK=Q4_01
replace Q4_01_DK=0.5 if Q4_DV2_ranking==6
tab Q4_01_DK Q4_01, miss

gen Q7_01_DK=Q7_01
replace Q7_01_DK=0.5 if Q7_DV3_ranking==6
tab Q7_01_DK Q7_01, miss

** Testing H1a, H1b, and H1c **
* Model 1: Testing H1a (The state of HR/Environment in Israel is good) ** 
reg Q3_01_DK i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableE7.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Testing H1b (Israeli Gov. should do more to improve the HR/Environment situation)
reg Q4_01_DK i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE7.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Testing H1c (Willingness to sign a petition calling the Israeli government to act to improve HR/Environment)
reg Q7_01_DK i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE7.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Testing H2 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01_DK i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableE8.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01_DK i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE8.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01_DK i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE8.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 


** Testing H3 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01_DK i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableE9.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01_DK i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE9.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01_DK i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE9.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

drop Q3_01_DK-Q7_01_DK


** Examining DK responses by condition **
* First DV
recode Q3_DV1_ranking (1/5=0) (6=1), gen(Q3_DK)
tab Q3_DV1_ranking Q3_DK
tab Q3_DK HR_conditions, chi col /*p=.648*/
tab Q3_DK HR_exp_rank_high, chi col /*p=.178*/
tab Q3_DK HR_exp_ENV, chi col /*p=.639*/
tab Q3_DK HR_exp_group, chi col /*p=.437*/

* Second DV
recode Q4_DV2_ranking (1/5=0) (6=1), gen(Q4_DK)
tab Q4_DV2_ranking Q4_DK
tab Q4_DK HR_conditions, chi col /*p=.252*/
tab Q4_DK HR_exp_rank_high, chi col /*p=.997*/
tab Q4_DK HR_exp_ENV, chi col /*p=.064*/
tab Q4_DK HR_exp_group, chi col /*p=.027*/

* Third DV
recode Q7_DV3_ranking (1/5=0) (6=1), gen(Q7_DK)
tab Q7_DV3_ranking Q7_DK
tab Q7_DK HR_conditions, chi col /*p=.008*/
tab Q7_DK HR_exp_rank_high, chi col /*p=.480*/
tab Q7_DK HR_exp_ENV, chi col /*p=.000*/
tab Q7_DK HR_exp_group, chi col /*p=.742*/

** Examining DK responses by demographics **
* Ideology
tab Q3_DK ideo_bloc, chi col /*p=.648*/
tab Q4_DK ideo_bloc, chi col /*p=.178*/
tab Q7_DK ideo_bloc, chi col /*p=.639*/

* Gender
tab Q3_DK Q22_gender, chi col /*p=.648*/
tab Q4_DK Q22_gender, chi col /*p=.178*/
tab Q7_DK Q22_gender, chi col /*p=.639*/

* Jewish
tab Q3_DK jewish, chi col /*p=.648*/
tab Q4_DK jewish, chi col /*p=.178*/
tab Q7_DK jewish, chi col /*p=.639*/

drop Q3_DK Q4_DK Q7_DK



** Table E10 - testing H3, with two interactions of the "ranking source" factor **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc
outreg2 using TableE10.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_group ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE10.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_group ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	
outreg2 using TableE10.doc, append se dec(2) alpha (.02, .1) symbol (**, *)  



*** Testing heterogenous effects: Interactions with ideological blocs ***
** Table E11 - testing H1a, H1b, and H1c **
* Model 1: Testing H1a (The state of HR/Environment in Israel is good) ** 
reg Q3_01 i.HR_exp_rank_high##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity
outreg2 using TableE11.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity) ///
addtext(Demographic controls, "YES")  


* Model 2: Testing H1b (Israeli Gov. should do more to improve the HR/Environment situation)
reg Q4_01 i.HR_exp_rank_high##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE11.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  


* Model 3: Testing H1c (Willingness to sign a petition calling the Israeli government to act to improve HR/Environment)
reg Q7_01 i.HR_exp_rank_high##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE11.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  


** Table E12 - testing H2 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity
outreg2 using TableE12.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_ENV##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE12.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_ENV##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE12.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  


** Table E13 - testing H3 **
* Model 1: The state of HR/Environment in Israel is good ** 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group_US##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity
outreg2 using TableE13.doc, replace se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")   

* Model 2: Israeli Gov. should do more to improve the HR/Environment situation
reg Q4_01 i.HR_exp_rank_high##i.HR_exp_group_US##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE13.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  

* Model 3: Willingness to sign a petition calling the Israeli government to act to improve HR/Environment
reg Q7_01 i.HR_exp_rank_high##i.HR_exp_group_US##i.ideo_bloc ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity	
outreg2 using TableE13.doc, append se dec(2) alpha (.02, .1) symbol (**, *) ///
drop (ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity)  ///
addtext(Demographic controls, "YES")  



*** Testing H1 - controlling for the "prior attitudes" ***
gen priors_DV1=Q2_SATIS_HUMANRIGHTS if HR_exp_ENV==0
replace priors_DV1=Q2_SATIS_ENV if priors_DV1==. & HR_exp_ENV==1
recode priors_DV1 (6=.)
replace priors_DV1= (priors_DV1-1)/4
sum priors_DV1

** Table E14 - testing H1a, H2, and H3 - with the first DV (for which we have priors) **
* Model 1: Testing H1a *
reg Q3_01 i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc priors_DV1
outreg2 using TableE14.doc, replace se dec(2) alpha (.02, .1) symbol (**, *)

/*
* Model 1: Testing H1a - with priors X Environment interaction *
reg Q3_01 i.HR_exp_rank_high ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc priors_DV1 c.priors_DV1##i.HR_exp_ENV
outreg2 using TableE14.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
*/

* Model 2: Testing H2 * 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc priors_DV1
outreg2 using TableE14.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

/*
* Model 2: Testing H2 - with priors X Environment interaction *
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_ENV ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc c.priors_DV1##i.HR_exp_ENV
outreg2 using TableE14.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
*/

* Model 3: Testing H3 * 
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc	priors_DV1
outreg2 using TableE14.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 

/*
* Model 3: Testing H3 - with priors X Environment interaction *
reg Q3_01 i.HR_exp_rank_high##i.HR_exp_group_US ib2.age_groups i.Q22_gender i.college_educ i.jewish Q25_religiosity i.ideo_bloc c.priors_DV1##i.HR_exp_ENV
outreg2 using TableE14.doc, append se dec(2) alpha (.02, .1) symbol (**, *) 
*/
drop priors_DV1
